# $Id: PKGBUILD 79940 2012-11-15 00:36:52Z arodseth $
# Maintainer: Alexander Rødseth <rodseth@gmail.com>
# Contributor: Andrea Scarpino <andrea@archlinux.org>
# Contributor: Ryan Coyner <rcoyner@gmail.com>

pkgname=mod_wsgi
pkgver=3.4
pkgrel=2
pkgdesc='Python WSGI adapter module for Apache'
arch=('x86_64' 'i686')
url='http://www.modwsgi.org/'
license=('APACHE')
depends=('apache' 'python')
makedepends=('setconf')
install=mod_wsgi.install
source=("http://modwsgi.googlecode.com/files/$pkgname-$pkgver.tar.gz")
sha256sums=('ae85c98e9e146840ab3c3e4490e6774f9bef0f99b9f679fca786b2adb5b4b6e8')

build() {
  cd "$srcdir/$pkgbase-$pkgver"

  ./configure --prefix=/usr \
    --with-apxs=/usr/sbin/apxs \
    --with-python=/usr/bin/python
  setconf Makefile LDLIBS '-lpython3 -lpthread -ldl -lutil -lm'
  make
}

package() {
  cd "$srcdir/$pkgbase-$pkgver"

  make DESTDIR="$pkgdir" install
}

# vim:set ts=2 sw=2 et:
